MAPLE ASSIGNMENT 3,   MATH 266

In this assignment, MAPLE will be used to solve a large problem involving the method of undetermined coefficients for a high order linear ODE. First, an example will be worked in MAPLE to give you the tools you will need. Consider the initial value problem

$$\frac{d^4 y}{dx^4} + 2\,\frac{d^3 y}{dx^3} + 3\,\frac{d^2 y}{dx^2} + 2\,\frac{dy}{dx} + 2\,y = \cos(x)$$

with   y(0)=1, y'(0)=2, y"(0)=3, and y'''(0)=4. Type in the following commands:

```
> char_poly := r^4 + 2 * r^3 + 3 * r^2 + 2 * r + 2;
```

$$char\_poly := r^4 + 2\,r^3 + 3\,r^2 + 2\,r + 2$$

```
> factor( char_poly );
```

$$(r^2 + 1)\,(r^2 + 2\,r + 2)$$

```
> solve( char_poly=0, r );
```

$$I,\ -I,\ -1 + I,\ -1 - I$$

```
> y_homo := c_1 *cos(x) + c_2 *sin(x) + c_3 *exp(-x)*cos(x) + c_4 *exp(-x)*sin(x);
```

$$y\_homo := c\_1\cos(x) + c\_2\sin(x) + c\_3\,e^{(-x)}\cos(x) + c\_4\,e^{(-x)}\sin(x)$$

```
> y_p := x * (a_1*cos(x)  + a_2*sin(x));    # This is the correct FORM of y_p.
```

$$y\_p := x\,(a\_1\cos(x) + a\_2\sin(x))$$

```
> diff(y_p,  x$4 ) + 2 * diff(y_p,  x$3 ) +3 * diff(y_p,  x$2 ) +2 * diff(y_p, x)+ 2 * y_p;
```

$$\begin{aligned}&-2\,a\_1\sin(x) + 2\,a\_2\cos(x) + 3\,x\,(a\_1\cos(x) + a\_2\sin(x)) - 4\,a\_1\cos(x) - 4\,a\_2\sin(x)\\ &\quad + 2\,x\,(a\_1\sin(x) - a\_2\cos(x)) + 3\,x\,(-a\_1\cos(x) - a\_2\sin(x))\\ &\quad + 2\,x\,(-a\_1\sin(x) + a\_2\cos(x))\end{aligned}$$

```
> simplify(");   # simplify previous expression, equate coefficients by hand
```

$$-2\,a\_1\sin(x) + 2\,a\_2\cos(x) - 4\,a\_1\cos(x) - 4\,a\_2\sin(x)$$

```
> eqn1 :=     -2*a_1  - 4*a_2 = 0;
```

$$eqn1 := -2\,a\_1 - 4\,a\_2 = 0$$

```
> eqn2 :=       2*a_2  - 4*a_1 = 1;
```

$$eqn2 := 2\,a\_2 - 4\,a\_1 = 1$$

```
> solve( {eqn1,eqn2} , {a_1,a_2} );
```

$$\left\{a\_2 = \frac{1}{10},\ a\_1 = \frac{-1}{5}\right\}$$

```
> assign(");      # this makes a_1=-1/5 and a_2=1/10 from now on
> y_gen := y_p + y_homo;   # y_gen is the general solution.
```

$$y\_gen := x\left(-\frac{1}{5}\cos(x) + \frac{1}{10}\sin(x)\right) + c\_1\cos(x) + c\_2\sin(x) + c\_3\,e^{(-x)}\cos(x) + c\_4\,e^{(-x)}\sin(x)$$

> subs(x=0, y_gen);

$$c\_1\cos(0) + c\_2\sin(0) + c\_3\,e^{0}\cos(0) + c\_4\,e^{0}\sin(0)$$

> value(");   # this causes the cos, sin, and e^0's above to get evaluated

$$c\_1 + c\_3$$

> EQN1 :=   1= " ; # the quote mark is shorthand for the previous expression.

$$EQN1 := 1 = c\_1 + c\_3$$

> subs(x=0 , diff(y_gen, x));
> value(");

$$-\frac{1}{5} + c\_2 - c\_3 + c\_4$$

> EQN2 :=   2 = " ;

$$EQN2 := 2 = -\frac{1}{5} + c\_2 - c\_3 + c\_4$$

> subs(x=0, diff(y_gen , x$2));
> value(");

$$\frac{1}{5} - c\_1 - 2\,c\_4$$

> EQN3 :=   3 = ";

$$EQN3 := 3 = \frac{1}{5} - c\_1 - 2\,c\_4$$

> subs(x=0, diff(y_gen , x$3));
> value(");

$$\frac{3}{5} - c\_2 + 2\,c\_3 + 2\,c\_4$$

> EQN4 :=   4 = ";

$$EQN4 := 4 = \frac{3}{5} - c\_2 + 2\,c\_3 + 2\,c\_4$$

> solve({EQN1, EQN2, EQN3, EQN4} , {c_1, c_2, c_3, c_4});

$$\left\{c\_2 = \frac{159}{25},\ c\_3 = \frac{113}{25},\ c\_1 = \frac{-88}{25},\ c\_4 = \frac{9}{25}\right\}$$

> assign(");
> y := y_gen;   # This is the solution to the IVP. (y_gen is no longer general)

$$y := x\left(-\frac{1}{5}\cos(x) + \frac{1}{10}\sin(x)\right) - \frac{88}{25}\cos(x) + \frac{159}{25}\sin(x) + \frac{113}{25}e^{(-x)}\cos(x) + \frac{9}{25}e^{(-x)}\sin(x)$$

```
> plot(y,x=0..10);
>
```



The assignment proper begins here. Use MAPLE as above to solve the problem,.

$$\frac{d^4 y}{dx^4} - \frac{d^3 y}{dx^3} - \frac{d^2 y}{dx^2} + \frac{dy}{dx} = x^2 + e^x$$

with y(0)=1, y'(0)=2, y''(0)=3, and y'''(0)=4. Plot the solution on an interval from x=-3 to x=2. Does the solution have a zero near x=-1?

Remarks: If you assign y:= 1/5 in a worksheet. then MAPLE will replace y by 1/5 from that point on. You can unassign this value by typing y := 'y' ;

I assigned a_1,a_2, c_1 through c_4, y_homo, y_p, and y_gen above. Rather than unassign all these, I could just select NEW from the file menue and start over.

The simplify command was useful above. You might also like the expand command.